Chapter 12 Communication Example Using Logistic Modeling

This document shows Mplus programming for the communication example in Chapter 12 using logistic-based SEM rather than probit-based SEM. I assume you have read Chapter 5 that introduces logistic regression and Chapter 12 and that you know Mplus syntax. Here I reproduce Figure 12.2 for the example and the core model equations:




FIGURE 1. Parent communication example

$$PA2 = a_1 + p_1 TREAT + b_1 BS1 + b_2 CQ1 + b_3 PA1 + d_1$$
 [1]

$$PK2 = a_2 + p_2 TREAT + b_4 BS1 + b_5 CQ1 + b_6 PK1 + d_2$$
 [2]

$$PE2 = a_3 + p_3 TREAT + b_7 BS1 + b_8 CQ1 + b_9 PE1 + d_3$$
 [3]

$$COM3 = a_4 + p_4 PA2 + p_5 PK2 + p_6 PE2 + p_7 TREAT + b_{10} BS1 + b_{11} CQ1$$
 [4]

LISEM ANALYSIS: THE LOGIT MODEL

In this section, I discuss how to apply the logit model in a LISEM context. I only focus on sections that require changing the probit syntax in Chapter 12 to logit syntax.

Total Effect of the Program on the Outcome

To evaluate the total effect of the program on parental communication I fit the logistic equation

$$COM3 = a + p_1 T + b_2 CQ1 + b_3 BS1$$
 [4]

but without mean centering making use of the MODEL CONSTRAINT feature in Mplus. The relevant code is in Table 1 and corresponds to Table 12.9 in Chapter 12.

Table 1: Syntax for Logit Profile Analysis for Total Effect with LISEM

```
1. TITLE: Logit-based total effect analysis of communication ;
2. DATA: FILE IS c:\mplus\communication.dat;
3. VARIABLE:
4. NAMES ARE ID COM3 PA2 PK2 PE2 CQ1 PA1 PK1 PE1 TREAT BS1;
5. USEVARIABLES ARE COM3 TREAT BS1 CQ1;
6. MISSING ARE ALL (-9999);
7. CATEGORICAL COM3;
8. ANALYSIS:
9. ESTIMATOR = ML;
10. MODEL:
11. COM3 ON TREAT BS1 CQ1 (p1 b1 b2);
12. [COM3$1] (thresh);
13. MODEL CONSTRAINT:
14. NEW(CLOGIT TLOGIT CPROB TPROB DIFF) ;
15. CLOGIT = -thresh + p1*0 + b1*.525 + b2*0.012 ;
16. TLOGIT = -thresh + p1*1 + b1*.525 + b2*0.012 ;
17. CPROB = EXP(CLOGIT)/(1 + EXP(CLOGIT));
18. TPROB = EXP(TLOGIT)/(1 + EXP(TLOGIT));
19. DIFF = TPROB-CPROB;
20. OUTPUT: SAMP STANDARDIZED (STDYX) RESIDUAL CINTERVAL TECH4;
```

In Line 9, I removed the command from Chapter 12 that read LINK=PROBIT. By specifying ML alone as the estimator, Mplus by default uses logistic modeling. On Line 14 in the NEW command, I changed the label names I used from CPROBIT and TPROBIT to CLOGIT and TLOGIT. Lines 17 and 18 changed the function to translates the logits into probabilities. The command EXP tells Mplus to take the exponent of the term in parentheses that follows it.

The Mplus out under MODEL RESULTS will now report logistic coefficients rather than probit coefficients. Mplus also reports the coefficients converted to odds ratios by taking the exponents of them. For example, here are the core model results for the logistic coefficients followed by the corresponding odds ratios:

MODEL RESULTS

		Estimate	S.E. Es	t./S.E.	Two-Tailed P-Value
COM3	ON				
TREAT		0.785	0.106	7.374	0.000
BS1		0.472	0.107	4.427	0.000
CQ1		0.344	0.139	2.481	0.013
Thresholds	3				
COM3\$1		0.775	0.096	8.084	0.000

LOGISTIC REGRESSION ODDS RATIO RESULTS

		Datimata	95% C.I. S.E. Lower 2.5% Upper 2.5%		
		Estimate	5.E. LO	ver 2.5%	opper 2.5%
COM3	ON				
TREAT		2.193	0.234	1.780	2.702
BS1		1.603	0.171	1.301	1.976
CQ1		1.410	0.195	1.075	1.851

The predicted odds of communication for parents in the treatment group are 2.193 times larger than the predicted odds of communication for parents in the control group, holding constant the covariates.

Here are the results from the MODEL CONSTRAINT commands:

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
New/Additional Parameters				
CLOGIT	-0.523	0.076	-6.907	0.000
TLOGIT	0.262	0.075	3.505	0.000
CPROB	0.372	0.018	21.028	0.000
TPROB	0.565	0.018	30.757	0.000
DIFF	0.193	0.026	7.562	0.000

The estimated proportion of parents who communicated with their children was 0.565 in the intervention group and 0.372 in the control group, with a proportion difference of 0.193, holding constant the covariates. It is straightforward to implement Harrell's recommendations to evaluate different profiles for total effects, as I did for the probit modeling.

Effects of the Mediators on the Outcome

I also used probit regression in Chapter 12 to evaluate if the targeted mediators are related to the outcome and to determine whether there are program effects on the outcome over and above the targeted mediators. Specifically, I worked with Equation 12.7, which I repeat here:

```
COM3 = a_4 + p_4 PA2 + p_5 PK2 + p_6 PE2 + p_7 TREAT + b_{10} BS1 + b_{11} CQ1
```

Table 2 presents the Mplus adapted syntax that maps onto Table 12.11 in Chapter 12.

Table 2: Syntax for Logit-Based Mediator Effects on Outcome

```
1. TITLE: Logit-based profile analysis
2. DATA: FILE IS c:\mplus\communication.dat;
3. VARIABLE:
4. NAMES ARE ID COM3 PA2 PK2 PE2 CQ1
5. PA1 PK1 PE1 TREAT BS1;
6. USEVARIABLES ARE COM3 PA2 PK2 PE2 CQ1 TREAT BS1 ;
7. CATEGORICAL ARE COM3;
8. MISSING ARE ALL (-9999);
9. ANALYSIS:
10. ESTIMATOR = ML ;
11. MODEL:
12. COM3 ON PA2 PK2 PE2 TREAT BS1 CQ1 (p4 p5 p6 p7 b10 b11) ;
13. [COM3$1] (thresh);
14. OUTPUT: SAMP STANDARDIZED (STDYX) CINTERVAL TECH4;
```

The only change I made was in Line 10 for the estimator so as to invoke logistic instead of probit estimation. The MODEL RESULTS section is the same but now logistic coefficients rather than probit coefficients are reported. As well, Mplus now provides odds ratios. Here is the output for the odds ratios:

LOGISTIC REGRESSION ODDS RATIO RESULTS

		Estimate	S.E.	959 Lower 2.5%	% C.I. Upper 2.5%
COM3	ON				
PA2		1.664	0.225	1.277	2.168
PK2		1.614	0.212	1.249	2.087
PE2		0.544	0.073	0.419	0.706
TREAT		0.997	0.182	0.698	1.426
BS1		1.478	0.162	1.193	1.832
CQ1		1.275	0.182	0.965	1.686

As an example of interpretation, for every one unit that PA2 increases, the predicted odds of parents communicated with one's child increases by a multiplicative factor of 1.664. You also can use the average marginal effects program on my website to calculate an average marginal effect for each mediator.

In Table 12.12 of Chapter 12, I presented syntax for a profile analysis strategy to evaluate the effects of mediators on outcomes. Table 3 presents the logistic version of the syntax.

Table 3: Syntax for Logit Model Profile Analysis with LISEM

```
1. TITLE: Logit-based profile analysis ;
2. DATA: FILE IS c:\mplus\communication.dat;
3. VARIABLE:
4. NAMES ARE ID COM3 PA2 PK2 PE2 CQ1
5. PA1 PK1 PE1 TREAT BS1;
6. USEVARIABLES ARE COM3 PA2 PK2 PE2 CQ1 TREAT BS1;
7. CATEGORICAL ARE COM3;
8. MISSING ARE ALL (-9999);
9. ANALYSIS:
10. ESTIMATOR = ML ;
11. MODEL:
12. COM3 ON PA2 PK2 PE2 TREAT BS1 CQ1 (p4 p5 p6 p7 b10 b11);
13. [COM3$1] (thresh);
14. MODEL CONSTRAINT:
15. NEW(CLOGIT TLOGIT CPROB TPROB DIFF);
16. CLOGIT = -\text{thresh}+p4*.469+p5*.433+p6*(-.059)+p7*.494+b10*.525+
17. b11*.012;
18. TLOGIT = -\text{thresh} + p4 \times 1.469 + p5 \times .433 + p6 \times (-.059) + p7 \times .494 + b10 \times .525 +
19. b11*.012;
20. CPROB = EXP(CLOGIT)/(1 + EXP(CLOGIT));
21. TPROB = EXP(TLOGIT)/(1 + EXP(TLOGIT));
22. DIFF = TPROB-CPROB ;
23. OUTPUT: SAMP STANDARDIZED(STDYX) CINTERVAL TECH4;
```

The only changes I made were to specify the ML estimator in Line 10 and the MODEL CONSTRAINT commands to reflect logit labels and use the appropriate transformation formula for the probabilities.

FISEM ANALYSIS: PROBIT

For the probit-based FISEM analysis, Table 4 presents the logistic syntax that maps onto the probit syntax in Table 12.13 of Chapter 12.

Table 4: Mplus Syntax for FISEM Logit-Based Model

```
1. TITLE: FISEM Logit analysis of RET
2. DATA: FILE IS c:\mplus\communication.dat;
3. DEFINE:
4. CENTER CQ1 BS1 PA1 PK1 PE1 (GRANDMEAN) ;
5. VARIABLE:
6. NAMES ARE ID COM3 PA2 PK2 PE2 CQ1
7. PA1 PK1 PE1 TREAT BS1;
8. USEVARIABLES ARE COM3 PA2 PK2 PE2 CQ1
9. PA1 PK1 PE1 TREAT BS1;
10. CATEGORICAL ARE COM3;
11. MISSING ARE ALL (-9999) ;
12. ANALYSIS:
13. ESTIMATOR = ML ;
14. MODEL:
15. PA2 ON BS1 CQ1 TREAT PA1 (b1 b2 p1 b3);
16. PK2 ON BS1 CQ1 TREAT PK1 (b4 b5 p2 b6);
17. PE2 ON BS1 CQ1 TREAT PE1 (b7 b8 p3 b9);
18. COM3 ON PA2 PK2 PE2 TREAT BS1 CQ1 (p4 p5 p6 p7 b10 b11) ;
19. [COM3$1] (thresh);
20. [PA2] (a1);
21. [PK2] (a2) ;
22. [PE2] (a3) ;
23. MODEL INDIRECT:
24. COM3 IND TREAT;
25. OUTPUT: SAMP STANDARDIZED (STDY) RESIDUAL CINTERVAL TECH4;
```

On Line 13, I changed the estimator to ML. I also removed the request for modification indices from the output line because Mplus does not compute them for logistic models of this type. Everything else remains the same. Mplus now reports logistic coefficients and their odds ratios. However, you will not obtain the more detailed global model fit statistics you obtain with probit analysis nor the modification indices, which is one reason I prefer the probit approach.

Total Effect of the Program on the Outcome

For logistic models, Mplus produces an estimate of the total effect using the latent propensity index, y*. Like probit-based analyses, you likely will want to use the STDY form of standardization when reporting this. For probabilities/proportions, you can use the same profile analysis strategy I described in Chapter 12, but instead of using PHI to translate probits into probabilities, you use the EXP coupled with the expressions described above when using the MODEL CONSTRAINTS command. For example, here are the probit commands I showed you in Chapter 12:

```
MODEL CONSTRAINT:
NEW (CPROBIT TPROBIT CPROB TPROB DIFF) ;
TPROBIT = -\text{thresh} + \text{p4} \cdot .885 + \text{p5} \cdot .859 + \text{p6} \cdot (-0.055) + \text{p7} \cdot 1 + \text{b10} \cdot 0 + \text{b11} \cdot 0;
CPROB = PHI(CPROBIT);
TPROB = PHI(TPROBIT);
DIFF = TPROB-CPROB ;
```

and here are the logistic counterparts:

```
MODEL CONSTRAINT:
NEW (CLOGIT TLOGIT CPROB TPROB DIFF) ;
CLOGIT = -thresh+p4*.063+p5*.017+p6*(-.062)+p7*0+b10*0+b11*0;
TLOGIT = -\text{thresh} + \text{p4} \cdot .885 + \text{p5} \cdot .859 + \text{p6} \cdot (-0.055) + \text{p7} \cdot 1 + \text{b10} \cdot 0 + \text{b11} \cdot 0;
CPROB = EXP(CLOGIT)/(1 + EXP(CLOGIT));
TPROB = EXP(TLOGIT)/(1 + EXP(TLOGIT));
DIFF = TPROB-CPROB ;
```

Effects of the Mediators on The Outcome

All of the methods I showed you in Chapter 12 to evaluate the effects of the mediators on the outcome carry over to the logistic model, with minor modifications that follow the same logic from above. The focus is on the following equation:

```
COM3 = a_4 + p_4 PA2 + p_5 PK2 + p_6 PE2 + p_7 T + b_{10} BS1 + b_{11} CQ1
```

You can use both average marginal effects Using the Appendix in Chapter 12) and profile analysis to execute the relevant analyses.